

## Defining a Style Template

### Basic Structure

```
proc template;
  define style style-name;
    ... statements / attributes ...
  end;
run;
```

### Commonly Used Statements

**style *style-element* < from *parent-style-element* > / *style-attributes*;**

Defines a style element with name *style-element* and attributes *style-attributes*. If specified, the attributes from *parent-style-element* are also included.

Attributes of the same name specified in the STYLE statement will override those in the parent. Multiple style elements can be defined simultaneously, by giving a comma-delimited list of names in *style-element*.

**class *style-element* / *style-attributes*;**

Defines a style element with the name *style-element* and parent of *style-element*. This is a shortcut for “style *style-element* from *style-element* / *style-attributes*;”. Multiple style elements can be defined simultaneously, by giving a comma delimited list of names in *style-element*.

**import “*URL*”;**

Reads a CSS file into the current style definition. *URL* can be either a local file path or a location on the Internet using the FTP or HTTP protocols.

### Commonly Used Attributes

**notes “*text*”;**

Specifies comments about the style definition that are stored with the template.

**parent=*style-definition*;**

Specifies a style definition to inherit style elements from.

## Style Definition Example

```
proc template;
  define style mystyle;
    notes "My Simple Style";
    class body /
      backgroundcolor = white
      color = black
      fontfamily = "Palatino"
    ;
    class systemtitle /
      fontfamily = "Verdana, Arial"
      fontsize = 16pt
      fontweight = bold
    ;
    class table /
      backgroundcolor = #f0f0f0
      bordercolor = black
      borderstyle = solid
      borderwidth = 1pt
      cellpadding = 5pt
      cellspacing = 0pt
      frame = void
      rules = groups
    ;
    class header, footer /
      backgroundcolor = #c0c0c0
      fontfamily = "Verdana, Arial"
      fontweight = bold
    ;
    class data /
      fontfamily = "Palatino"
    ;
  end;
run;

ods pdf style=mystyle;
proc report data=sashelp.class;
run;
ods pdf close;
```

## The SAS System

Name	Sex	Age	Height	Weight
Alfred	M	14	69	112.5
Alice	F	13	56.5	84
Barbara	F	13	65.3	98
Carol	F	14	62.9	102.5



# PROC TEMPLATE Styles Tip Sheet

This tip sheet places frequently used information in one place, on one sheet of paper, so you don't have to search through the online documentation. It also gives you something to take home, type in, and try.

PROC TEMPLATE styles give you the power to customize the look and feel of your ODS reports. You can change the fonts, colors, borders, and add images and text. It is also possible to set the style of data cells in tables based on the data itself. This tip sheet presents the most common statements and attributes used in creating styles with PROC TEMPLATE.



SAS Institute Inc. World Headquarters  
+1 919 677 8000 [www.sas.com/offices](http://www.sas.com/offices)

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies. Copyright © 2008, SAS Institute Inc. All rights reserved. 479398\_01.08

# SAS<sup>®</sup>9 PROC TEMPLATE Styles Tip Sheet

## Terms

### style attribute

A name-value pair that describes a single behavioral or visual aspect of a piece of output.

### style element

A named collection of style attributes specified by the STYLE or CLASS statement. Each area in a piece of ODS output has a style element name associated with it where the attributes will be applied.

### style | style template | style definition

All three terms are commonly used to describe a named collection of style elements. The style definition name is used in the STYLE= option of the ODS statement.

### Cascading Style Sheets (CSS)

An alternative style language to PROC TEMPLATE styles written by the World Wide Web Consortium (W3C) which is commonly used to apply styles to web pages. For more information, see <http://www.w3.org/Style/CSS/>.

## Browsing Templates

The style templates supplied by SAS<sup>®</sup> are stored in the SASHELP.TMPLMST itemstore. You can browse the templates using either a graphical interface or PROC TEMPLATE code.

### odstemplates

Type **odstemplates** into the command bar to open the Template Browser. Double-clicking a template displays its source code. The styles supplied by SAS are located in the styles directory of the SASHELP.TMPLMST itemstore.

### proc template; list styles; run;

Lists all templates in **styles** directory. This is the default location for the styles supplied by SAS.

### proc template; source style-name; run;

Displays the source of **style-name** to the log.

## Using a Style Definition

The following code applies a style with the name *style-name* to an ODS destination.

```
ods destination style=style-definition ... ;
```

All ODS destinations except for document, listing, and output support the STYLE= option.

## Using a CSS File

The following code loads a CSS file rather than a PROC TEMPLATE style.

```
ods destination cssstyle="URL" ... ;
```

where *URL* is a local file path or a location on the Internet using the FTP or HTTP protocols.

All ODS destinations except for document, listing, and output support the CSSSTYLE= option.

## Style Attribute Values

### dimension

A positive integer or floating point value followed by one of the following units:

**pt** – points (1/72 inch)  
**in** – inches  
**cm** – centimeters (2.54 cm = 1 inch)  
**mm** – millimeters (25.4 mm = 1 inch)  
**%** - percentage (percentage of current font size for fontsize=, percentage of container width or height for width= and height=)  
**ex** – height of the ‘x’ character  
**em** – width of the ‘m’ character  
**px** – pixels

### color-name

Any predefined SAS/GRAPH<sup>®</sup> color name.

### hex-color

Indicates the red, green, and blue components of a color (from 0 to 255) in hexadecimal preceded by the # character. For example, red is #ff0000, green is #00ff00, and blue is #0000ff.

## Commonly Used Style Attributes

### General Style Attributes

**backgroundcolor=***color-name* | *hex-color*  
Specifies the background color of a region.

**backgroundimage=**“*path-to-file*”  
Specifies an image to put in the background.

**color=***color-name* | *hex-color*  
Specifies the color of the text content.

**height=***dimension*  
**width=***dimension*  
Specifies the width and height of an element.

**preimage=**“*path-to-file*”  
**postimage=**“*path-to-file*”  
Specifies an image to put before/after the content.

**pretext=**“*text*”  
**posttext=**“*text*”  
Specifies text to put before/after the content.

### Font and Text Style Attributes

**fontfamily=**“*fontfamily-1, fontfamily-2, ...*”  
Specifies a list of font family alternatives for the text content. The first usable font found on the system will be used.

**fontsize=***dimension*  
Specifies the size of the text.

**fontstyle=***italic* | *roman* | *slant*  
Specifies the style of the text.

**fontweight=***bold* | *medium*  
Specifies whether the text should be bold or not.

**textdecoration=***line\_through* | *overline* | *underline*  
Indicates that a horizontal line should be placed through, over, or under the text.

## Commonly Used Style Attributes

(Continued)

### Table Style Attributes

**borderspacing=***dimension*  
**cellspacing=***dimension*  
Specifies the amount of space to put between adjacent table cells.

**frame=***box* | *above* | *below* | *hsides* | *vsides* | *lhs* | *rhs* | *void*  
Specifies which borders should appear around the perimeter of the table.

**padding=***dimension*  
**cellpadding=***dimension*  
Specifies the amount of space to put between the content and border of the cells.

**rules=***all* | *cols* | *rows* | *groups* | *none*  
Specifies which borders should appear within the table.

### Border Style Attributes

**bordercolor=***color-name* | *hex-color*  
Specifies the color of the border on all sides. Individual borders can be set using **bordercolor=**, **borderbottomcolor=**, **borderleftcolor=**, **borderrightcolor=**, and **borderstyle=**.

**borderstyle=***dashed* | *dotted* | *double* | *groove* | *hidden* | *inset* | *outset* | *ridge* | *solid* | *none*  
Specifies the style of the border on all sides. Individual borders can be set using **borderstyle=**, **borderbottomstyle=**, **borderleftstyle=**, **borderrightstyle=**, and **borderstyle=**.

**borderwidth=***dimension*  
Specifies the width of the border on all sides. Individual borders can be set using **borderwidth=**, **borderbottomwidth=**, **borderleftwidth=**, **borderrightwidth=**, and **borderwidth=**.